CHAPTER 8
MAKING SOUND AND MUSIC ON THE PLUS/4

• Introduction
• Volume command
• Sound command
• Creating sound effects
• Making some music
• The Plus/4 Music Machine
INTRODUCTION

Here is a short program to make music on your Plus/4. Type in the program exactly as it appears, and remember to press [RETURN] at the end of each line. When the program is typed in, type RUN and then press [RETURN]. When the program asks you to enter a number, type any number from 0 to 1023 and press [RETURN]. To stop the program, enter a zero as your value.

```
10 VOL 8
20 DO
30 INPUT X
35 IF X > 1023 or X<0 THEN PRINT "0 TO 1023, PLEASE": GO TO 30
40 SOUND 1, X, 10
50 LOOP UNTIL X = 0
```

Press the RUN/STOP key to stop the program.

Here's how to play a single note on your Plus/4. Type NEW and press [RETURN] to clear the Plus/4's memory.

First: Type NEW and press [RETURN]
Type VOL 8 and press [RETURN]

Second: Type SOUND 1,266,60 and press [RETURN]

You should hear a note play for about a second and then stop. If you don't hear anything, turn up the volume of your television or monitor and try it again.

These two steps are the only commands that you need to know to play music on your Plus/4. Let's look at what these two commands do.

THE VOLUME COMMAND

The VOL command controls the VOLUME of the notes that the Plus/4 plays. Think of the first three letters of the word "volume" to remember the VOL command. The number that comes after VOL is the setting for the volume. Think of the VOL command as a volume knob on the Plus/4. When the knob points to zero, the volume is off and you won't hear anything. When the knob is set at 8, the volume is turned up all the way, and your Plus/4 plays as loud as it can.

Try the first example again and use a different number after the VOL command. The smaller the number, the softer the note is played.
The SOUND command tells your Plus/4 everything it needs to know about the sound you want to play. The SOUND command is followed by three numbers that describe the note:

SOUND voice, note value, duration

The first number in the sound command refers to voice. The number for voice can be a 1, 2 or 3. The Plus/4 sound is produced by two different “voices”, 1 for the first voice and 2 for the second voice. The third voice option applies to voice 2’s ability to produce either a tone or noise.

Voice 1—This voice plays only tones. Select this voice with a 1 after the SOUND command.

Voice 2—This voice is like voice 1, but can be used to play tones or noise for sounds. Type a 2 in the command to use this voice for tones, or a 3 to use this voice for noise, to make sound effects like thunder and rain.

The second number after the word SOUND is the note value (frequency). This can be any number from 0 to 1023. It tells your Plus/4 how low- or high-pitched a note to play. As the numbers get larger, the notes get higher. The highest values (in the 1023 neighborhood) are not audible to the human ear.

Note: Noise is “white” only in the range of 600-940 with Voice 3. You can use register values outside this range to create interesting sound effects.

Here is a chart that shows all of the notes in one scale, along with the note value to use. There is a complete chart of notes for the Plus/4 in Section 11 of the Plus/4 Encyclopedia.
Try the following program on your Plus/4:

```
NEW
10 VOL 7  sets VOLume
20 X = 0
30 DO
40 SOUND 1,X,5  plays note
50 X = X + 5
60 LOOP UNTIL X = 1020
70 VOL 0  turns off VOLume
80 END
```

TYPE RUN AND PRESS Enter. This program allows your Plus/4 to show off some of its musical range.

The third number after the word SOUND controls the duration (length) of the note. This tells the Plus/4 how long to play the note. This number can be anything from 0 to 65535. This number sets a timer, which counts time in sixtieths of a second. A duration of 60 keeps the note on for one second. A rule of thumb for duration is the larger the number, the longer the note stays on. In fact, if you use 65535, the note stays on for over 16 minutes. To turn a sound off, use a zero duration, which does not allow the sound to be produced.

Sound effects can be created using either musical tones or noise. Combining simple BASIC programs and sound commands can generate unusual and entertaining effects. For instance, the FOR... NEXT... STEP loop can be used creatively in sound effects. This program uses a FOR... NEXT loop with a negative step, to count down from a high number to a lower one.

```
10 VOL 8  sets VOLume at 8
20 FOR S = 1000 TO 700 STEP - 25
30 SOUND 1, S, 1
40 NEXT S
```

Type RUN and press Enter to hear the sound effect. The key is line 20, which selects a number range from 1000 to 700 going down the scale. STEPping down 25 numbers at a time. Finally, line 30 instructs your Plus/4 to play each note for just an instant by setting the DURATION to 1, which is 1/60 of a second. Experimenting with different number and duration values can give you some very interesting effects.
Creating A Noise Sound Effect

Using a value of 3 when selecting a voice in the SOUND command specifies noise. This is used to create sound effects with noise rather than tone. The following program uses voice 3 to create the sounds of a windstorm.

```
10 VOL 2  
20 R = INT (RND(0)*10) + 1  
30 FOR X = 1 TO R  
40 SOUND 3, 600 + 30*X, 10  
50 NEXT X  
60 FOR X = R TO 1 STEP - 1  
70 SOUND, 3,600 + 30*X, 10  
80 NEXT X  
90 T = INT (RND(0)*100) + 30  
100 SOUND 3, 600, T  
110 GOTO 20
```

Lines 30 and 60 set up loops for the note value (frequency) of the sound, one increasing and one decreasing, based on the random number from line 20. It is important to have variation in pitch, since windstorms have different forces of gusts of wind. Lines 40 and 70 are the SOUND commands that create the noise. Lines 90 and 100 set up a random delay to recreate the uneven nature of a windstorm with time lapses between howls. The program selects a Random number that is used for the duration of another SOUND command. This SOUND command stays at the same pitch and provides a consistent background noise that serves as a counterpoint to the gusts of wind.

Creating sound effects using noise is extremely challenging, trying to capture the right elements of the sound you want exactly. To create good sound effects, you have to be willing to experiment.

Making Some Music

Now that we've looked at some ways to create sound effects on your Plus/4, let's make some music. Here are a couple of programs to try:

The first program turns the keys from 1 through 8 into a piano. Type in the program and then type RUN.

```
5 SCNCLR  
10 FOR X = 1 TO 8: READ N(X): NEXT X  
20 VOL 7  
30 DO  
40 GET A$: IF A$ = " " THEN 40  
50 A = ASC(A$): A<49 OR A>56 THEN 90  
60 N = A - 48  
70 SOUND 1, N(N), 5
```
80 COLOR 0, N, 3   
90 LOOP UNTIL A = 32   
100 VOL 0: COLOR 4, 2, 7   
110 DATA 169, 262, 345, 383, 453, 516, 571, 596

Press numbers 1 through 8 to play notes. The screen border even changes colors with the different notes. When you finish playing, press the space bar to stop the program.

Here are the numbers to press for a familiar song:

**Twinkle, Twinkle Little Star**

1 1 5 5 6 6 5
4 4 3 3 2 2 1
5 5 4 4 3 3 2
5 5 4 4 3 3 2
1 1 5 5 6 6 5
4 4 3 3 2 2 1

![Musical notes](image)

This program plays a song by reading a list of DATA statements. The DATA statements are in pairs. The first number is the note value for the SOUND command and the second number is the duration for the SOUND command.

**Row Boat**

10 VOL 8  
20 DO  
30 READ X, Y  
40 SOUND 1, X, Y  
45 FOR D = 1 to 550:NEXT  
50 LOOP UNTIL X = 0  
60 END  
100 DATA 169, 45, 169, 45, 169, 30  
110 DATA 169, 15, 345, 45, 345, 45, 345, 30  
120 DATA 262, 15, 345, 30, 383, 15  
130 DATA 453, 60, 596, 45, 453, 45  
140 DATA 345, 45, 169, 45, 453, 30  
150 DATA 383, 15, 345, 30, 262, 15  
200 DATA 0, 0

This loop creates a brief delay between notes.
This program plays notes going up and down scales at different speeds, and displays some color bars along with them.

```
10 VOL 8
20 DO
30 D = INT(RND(0)*5) + 2: REM DURATION
40 S = INT(RND(0)*300) + 700: REM START
50 R = INT(RND(0)*(1020 - S)): REM RANGE
60 P = INT(RND(0)*30) + 5: REM STEP
70 T = SGN (RND) (1) - .5: IF T = 0 THEN 70
80 FOR Z = S TO S + T*R STEP P*T
90 SOUND 1, Z, D
100 Y = (Z AND 15) + 1: FOR X = 1 TO D
110 PRINT CHR$(18);: COLOR 1, Y: PRINT " " ;
120 NEXT X, Z
130 LOOP
```

THE GREAT PLUS/4 MUSIC MACHINE

The last program is a little longer. This is the "GREAT PLUS/4 MUSIC MACHINE". When you press a key from 1 through 9, the note is played, and a note appears on the staff on the correct line.

```
5 GOSUB 1000
6 FOR X = 1 TO 9: READ N(X): NEXT X
8 CHAR 1, 8, 1, " THE GREAT MUSIC MACHINE"
10 VOL 7
20 DO
30 GET A$: IF A$ = "" THEN 30
35 A = ASC(A$): A < OR A > 57 THEN 50
36 N = A - 48
40 SOUND 1, N(N), 4
45 GSHAPE N$, 150, 8 * (6 + (9 - N)), 4
46 FOR Z = 1 TO 50: NEXT Z
47 GSHAPE N$, 150, 8 * (6 + (9 - N)), 4
50 LOOP UNTIL A = 32
55 VOL 0: GRAPHIC 0: SCNCLR
60 END
100 DATA 345, 383, 453, 516, 596, 643, 685, 704
1000 GRAPHIC 1,1
1010 FOR Y = 60 TO 124 STEP 16
1020 DRAW 1, 100, Y TO 200, Y
1030 NEXT Y
1040 A$ = "FEDCBAGEF"
1050 FOR X = 1 TO 9: C = 13
1060 IF INT(X/2) = X/2 THEN C = 14
1070 CHAR 1, C, X + 6, MID$ (A$, X, 1) 0
1075 CHAR 1, C + 10, X + 6, RIGHT$ (STR$ (10 - X), 1)
1080 NEXT X
```
As you can see, it's not hard to write your own sound programs. The ones in this chapter just give a taste of the music capabilities of your Plus/4. Don't be afraid to try new sounds and noises and create your own masterpiece.